VBA Code Sample #61 – Creating an Outlook Distribution List from an Excel Worksheet
Created 

Helen Feddema (hfeddema@hvc.rr.com, http://www.helenfeddema.com)

Office version:  2003 and up
Instructions

This code creates an Outlook Distribution List from a list of names and email addresses in an Excel worksheet.
To test the code sample, follow the steps below:

1. Decide which folder you want to use to store the names from Excel (it must be a Contacts folder, and have Show this folder as an e-mail Address Book checked in its Properties sheet).  In the sample it is called DL Contacts.

2. Open the Outlook Address Book dialog (Tools|Address Book), and select Tools|Options.

3. Select the DL Contacts folder in the Show this address list first drop-down list, and click OK.

4. Open the Outlook VB project (select Tools|Macro|Visual Basic Editor from the main Outlook window, or click Alt-F11).
5. Select File|Import and browse for the DL.bas file to import it into your VB project.

6. Close the VB window and in the main Outlook window, right-click the grey background of the menu bar and select Customize.

7. In the Customize dialog, select Macros in the Categories list, and locate the CreateDL procedure in the Commands list.

8. Drag the CreateDL procedure to the Outlook toolbar.

9. Right-click the new button, and edit its name as desired.

10. Now you can click on the button to run the code that creates the distribution list from the Excel worksheet.

11. As the code runs, you will be asked to select the Outlook folder to use for the names:

12. And you will get an InputBox where you can enter the name of the Outlook Distribution List to create:

13. And another asking for the name and path of the Distribution List workbook:

14. The code will cycle through the rows in the worksheet, make a contact item for each one in the designated Contacts folder, and then a Distribution list member, and finally open the Distribution List.  The figure below shows the Outlook Distribution list and the Excel worksheet from which it was created:

Code

Public Sub CreateDL()

'Created by Helen Feddema 12-Aug-2007

'Last modified 12-Aug-2007

On Error GoTo ErrorHandler

   Dim appExcel As New Excel.Application

   Dim con As Outlook.ContactItem

   Dim DL As Outlook.DistListItem

   Dim fil As Scripting.File

   Dim fldContacts As Outlook.MAPIFolder

   Dim fso As New Scripting.FileSystemObject

   Dim lngCurrentRow As Long

   Dim lngRows As Long

   Dim nms As Outlook.NameSpace

   Dim rcp As Outlook.Recipient

   Dim rngEMail As Excel.range

   Dim rngName As Excel.range

   Dim strDLName As String

   Dim strEMail As String

   Dim strName As String

   Dim strPath As String

   Dim strPrompt As String

   Dim strRange As String

   Dim strTitle As String

   Dim strWorkbook As String

   Dim strWorkbookPath As String

   Dim wkb As Excel.Workbook

   Dim wks As Excel.Worksheet

   Set nms = Application.GetNamespace("MAPI")

SelectFolder:

   Set fldContacts = nms.PickFolder

   If fldContacts.DefaultItemType <> olContactItem Then

      strTitle = "Wrong folder type"

      strPrompt = "Please select a Contacts folder"

      MsgBox prompt:=strPrompt, _

         Buttons:=vbExclamation + vbOKOnly, _

         Title:=strTitle

      GoTo SelectFolder

   End If

   Set DL = Application.CreateItem(olDistributionListItem)

   'Get Distribution List name from InputBox

   strDLName = InputBox(prompt:="Distribution List name:", _

      Title:="DL name", _

      Default:="Excel contacts")

   If strDLName = "" Then

      GoTo ErrorHandlerExit

   End If

   DL.DLName = strDLName

   'Get worksheet name and path from inputbox

   '(FileDialog doesn't work from Outlook)

   strWorkbook = InputBox(prompt:="Workbook name and path:", _

      Title:="DL workbook", _

      Default:="K:\Documents\Web Site\Code Sample 61\Distribution List.xls")

   If strWorkbook = "" Then

      GoTo ErrorHandlerExit

   Else

      strWorkbookPath = Left(strWorkbook, Len(strWorkbook) - 21)

On Error Resume Next

      'Check that workbook is in the specified location

      Set fil = fso.GetFile(strWorkbook)

      If fil Is Nothing Then

         strPrompt = "Can't find Distribution List workbook " _

            & " in " & strWorkbookPath & "; canceling"

         MsgBox strPrompt, vbCritical + vbOKOnly

         GoTo ErrorHandlerExit

      End If

On Error GoTo ErrorHandler

   End If

   'Or can hard-code the name and path:

   'strWorkbook = "K:\Documents\Web Site\Code Sample 61\Distribution List.xls"

   'Open workbook and activate worksheet

   Set wkb = appExcel.Workbooks.Open(strWorkbook)

   Set wks = wkb.Sheets(1)

   wks.Activate

   appExcel.Visible = True

   'Get last data row number

   lngRows = wks.UsedRange.Rows.Count

   'Set up loop to process names and add to DL

   strRange = "A1"

   Set rngName = appExcel.range(strRange)

   rngName.Select

   Do While lngCurrentRow <= lngRows

      lngCurrentRow = rngName.Row

      'Debug.Print "On Row " & lngCurrentRow

      strName = ""

      strEMail = ""

      strName = rngName.Value

      If strName = "" Then

         GoTo NextRow

      End If

      Set rngEMail = appExcel.ActiveCell.Offset(columnoffset:=1)

      strEMail = rngEMail.Value

On Error Resume Next

      'Attempt to set reference to contact, in case it

      'already exists

      Set con = fldContacts.Items(strName)

On Error GoTo ErrorHandler

      If con Is Nothing Then

         'Create Contact item for name

         Set con = fldContacts.Items.Add(olContactItem)

         con.FullName = strName

         con.FileAs = strName

         con.Email1Address = strEMail

         con.Close (olSave)

      End If

      Set con = Nothing

      Set rcp = Application.Session.CreateRecipient(strName)

      rcp.Resolve

      If rcp.Resolved = True Then

         DL.AddMember rcp

      Else

         strTitle = "Problem with recipient"

         strPrompt = "Recipient " & strName & _

            " could not be resolved; skipping"

         MsgBox prompt:=strPrompt, _

            Buttons:=vbExclamation + vbOKOnly, _

            Title:=strTitle

      End If

NextRow:

      rngName.Select

      Set rngName = appExcel.ActiveCell.Offset(rowoffset:=1)

      rngName.Select

   Loop

   DL.Save

   DL.Display

ErrorHandlerExit:

   Set appExcel = Nothing

   Exit Sub

ErrorHandler:

   MsgBox "Error No: " & Err.Number & "; Description: " & _

      Err.Description

   Resume ErrorHandlerExit

End Sub
References

The code in the sample module needs the following references (in addition to the default references):

Microsoft Scripting Runtime

Microsoft Excel 11.0 Object Library

If you import code or objects into a database of your own, you may need to set one or more of these references.  The version number may differ, depending on your Office version; check the version you have.  References are set in the References dialog, opened from the VBA window.  For more information on working with references, see Access Archon #107, Working with References
Notes

· The code uses an address list called DL Contacts (actually, a Contacts folder designated as an address list).  You must make this address list the first one selected in the Address Lists dialog for the code to work; there does not seem to be any way to do this programmatically.

· Contact items have to be created first, so the names and email addresses can be added to a Distribution List (otherwise they won't resolve).

· This code will probably work in Outlook 2000, 2002 and 2007, but I have only tested it in Outlook 2003.
Contents of Zip File

	File Name
	File Type
	Destination Folder

	DL.xls
	Excel 97-2003 worksheet
	Wherever you want

	DL.bas
	Saved VBA module
	Import into your Outlook VB project

	VB Script Code Sample #61 – Creating an Outlook Distribution List from an Excel Worksheet
	Word 97-2003 document
	\My Documents\Outlook (or wherever you want)

